Scand. Actuarial J. 1989: 119–122

SAJ WORKSHOP

Numerical calculation of the Cramér-Lundberg approximation

By Stig Rosenlund

We consider the classical risk process, i.e. the claims occur in a Poisson process with intensity λ and are independent with common distribution F, where $F(0)=0$, and mean μ. The premium rate is c.

The purpose of this note is to study the numerical estimation, from empirical data, of the Cramér-Lundberg approximation of the ruin probability. We shall demonstrate a quite simple algorithm demanding very little computer time even for a very large claim sample.

Define

$\varrho = (c-\lambda\mu)/\lambda\mu$, assumed >0

$$g(r) = \int_0^\infty e^{rz} dF(z) - 1 - cr/\lambda$$

$\psi(u) =$ probability of eventual ruin with initial capital u.

We assume $g(r)<\infty$ for some $r>0$. Thus F cannot be too heavy-tailed, for instance lognormal or Pareto.

Let R be the positive solution r of $g(r)=0$. Then the Cramér-Lundberg approximation is

$$\psi(u) \sim [\varrho\mu/g'(R)] \exp(-Ru) \quad (u \to \infty) \tag{1}$$

See Cramér (1930) and, for a modern presentation, Feller (1971).

This note describes a numerical solution R^* to the equation $G(r)=0$, where

$$G(r) = \int_0^\infty e^{rz} dF^*(z) - 1 - cr/\lambda^*$$

and F^* is the empirical distribution function and λ^* is the estimated Poisson intensity, formed from a large number of claims in a sample realization of the risk process in finite time.

Substituting corresponding functions of the sample for ϱ, μ, g' and R^* for R in (1) gives a natural estimate of the ruin probability for large u. The

Lundberg inequality $\psi(u) \leq \exp(-Ru)$ for all u also gives a motive for finding an estimate of R. For this inequality see Lundberg (1926) and Gerber (1979).

See Grandell (1979) for a treatment of estimation of ruin probabilities.

Classical ruin theory cannot of course be applied directly to a real insurance business. Still, it can be valuable by providing a lower bound for the probability of ruin and by providing a building block for a larger model, where the stochastic fluctuations of λ, F, c and of the financial business are taken into account.

In the sample realization we had 182342 claims, occurring during one year in the LF group. We used the secant method for finding the positive zero of G, which for this case is more efficient than the Newton-Raphson method. Two initial values r_0, r_1 are to be given, and then successive approximations are obtained from

$$r_{n+2} = r_{n+1} - G(r_{n+1})[r_{n+1} - r_n]/[G(r_{n+1}) - G(r_n)], \; n \geq 0 \tag{2}$$

Convergence of r_n to R^* is guaranteed if we have found $0 \leq \tau_1 < \tau_2 < \tau_3$ such that $G(\tau_1) < G(\tau_2) < G(\tau_3)$ and take $r_0 = \tau_2$, $r_1 = \tau_3$. Then G is increasing and convex on (r_0, ∞).

A preliminary program gives $G(r)$ for a few values of r in order to find r_0 and r_1.

Then a main program takes these as initial values and iterates according to (2) until sufficient accuracy is obtained. We chose the stopping rule

$$|r_{n+2} - r_{n+1}|/r_{n+2} < 10^{-8}$$

which gives r_{n+2} as a sufficient approximation to R^*.

On the mainframe computer IBM 3090, using the VS BASIC language in TSO under MVS/XA, computing $G(r)$ for 182342 claims took about 1 CPU-second. Eight calculations of $G(r)$, two for initial values and six iterations, were needed. Together with one CPU-second for computing the derivative G' and about five CPU-seconds for reading all claims into memory the total CPU time was about 14 seconds.

The limitation in this algorithm is in how much internal memory can be used for the claims. For the computer and program language used here about 870000 claims can be read into memory. For a larger number of claims these have to be read from disk at every computation of $G(r)$, which significantly increases CPU time.

VS BASIC is the most powerful and versatile mathematical language of those available at the LF group. However, with a still more powerful mathematically oriented third generation language like PL/I (not installed at LF) running on a more powerful computer, several millions of claims can be read into memory.

In the VS BASIC language an array can have a maximum of 32767 elements, hence several arrays were used to store the claims. The program listed below thus looks more complicated than it really is.

<u>Main program</u>

```
10000 OPTION LPREC
10100 REM CLAIMS IN THE LF-GROUP INCURRED 1987
10200 DIM A(32000),B(32000),C(32000),D(32000),E(32000),F(22342)
10300 N=182342              REM NUMBER OF CLAIMS, ESTIMATE OF λ
10400 C=1833750000          REM ASSUMED PREMIUM RATE
10500 U1=9145.02            REM AVERAGE CLAIM, ESTIMATE OF μ
10600 S1=9.9686143188E-02 REM ESTIMATE OF ρ
10700 DEF FNG(R)            REM EMPIRICAL FUNCTION G(R)
10800   G=0
10900   FOR J=1 TO 32000
11000   G=G+EXP(R*A(J))+EXP(R*B(J))+EXP(R*C(J))+EXP(R*D(J))+EXP(R*E(J))
11100   NEXT J
11200   FOR J=1 TO 22342
11300   G=G+EXP(R*F(J))
11400   NEXT J
11500   RETURN G/N - 1 - R*C/N
11600 FNEND
11700 DEF FNH(R)            REM DERIVATIVE OF G(R)
11800   H=0
11900   FOR J=1 TO 32000
12000   H=H+A(J)*EXP(R*A(J))+B(J)*EXP(R*B(J))+C(J)*EXP(R*C(J))+D(J)*
        EXP(R*D(J))+E(J)*EXP(R*E(J))
12100   NEXT J
12200   FOR J=1 TO 22342
12300   H=H+F(J)*EXP(R*F(J))
12400   NEXT J
12500   RETURN H/N - C/N
12600 FNEND
12700 FOR J=1 TO 32000                      REM    CLAIMS READ
12800 GET 'RUIN',A(J),B(J),C(J),D(J),E(J)  REM    INTO ARRAYS
12900 NEXT J                                REM    IN MEMORY
13000 FOR J=1 TO 22342                      REM    FROM DISKFILE 'RUIN'
13100 GET 'RUIN',F(J)                       REM
13200 NEXT J                                REM
13300 R0=9E-8                               REM    STARTING VALUES
13400 R1=2E-7                               REM    FOR ABSCISSAS
13500 G0=FNG(R0)                            REM    STARTING VALUES
13600 G1=FNG(R1)                            REM    FOR ORDINATES
13700 R2=R1-G1*(R1-R0)/(G1-G0)              REM
13800 IF ABS(R2-R1)/R2 < 1E-8 THEN 14400   REM    SECANT METHOD
13900 G0=G1                                 REM    FOR DETERMINING
14000 G1=FNG(R2)                            REM    ZERO OF
14100 R0=R1                                 REM    FUNCTION G(R)
14200 R1=R2                                 REM
14300 GOTO 13700                            REM
14400 PRINT 'APPROXIMATE SOLUTION OF G(R)=0 IS ';R2
14500 PRINT 'ABS(R2-R1) IS ';ABS(R2-R1)
14600 PRINT 'G('!!CHR(R1)!!') IS ';G1
14700 C1=FNH(R2)
14800 PRINT 'ESTIMATE OF CRAMÉR-LUNDBERG APPROXIMATION OF ψ(U) IS'
14900 PRINT CHR(S1*U1/C1)!!'*EXP(-'!!CHR(R2)!!'*U)  (U → ∞).'
15000 END
```

References

Cramér, H. (1930). On the mathematical theory of risk. *Skandia Jubilee Volume*, Stockholm.

Feller, W. (1971). *An introduction to probability theory and its applications*, vol. II, 2nd ed. Wiley, New York.

Gerber, H. U. (1979). *An introduction to mathematical risk theory*. S. S. Heubner Foundation monograph series 8, Philadelphia.

Grandell, J. (1979). Empirical bounds for ruin probabilities. *Stochastic Proc. Applic.* **8**, 243–255.

Lundberg, F. (1926). *Försäkringsteknik riskutjämning*. F. Englunds boktryckeri A.B., Stockholm.

Received September 1988

Stig Rosenlund
Länsförsäkringsbolagens AB
S-103 81 Stockholm
Sweden